
***** Table 2 -- Impacts from Kenya Extension

** Impacts from first survey

forvalues i = 1/6{

gen e_rf_binary`i' = .

count if inlist(e_rf`i',1,2)
local num_support = r(N)
count if inlist(e_rf`i',4,5)
local num_oppose = r(N)

replace e_rf_binary`i' = e_rf`i' <= 3 if `num_support' < `num_oppose'
replace e_rf_binary`i' = e_rf`i' <= 2 if `num_support' >= `num_oppose'
}

swindex e_rf_binary1 e_rf_binary2 e_rf_binary3 e_rf_binary6, gen(e_rf_index) replace normby(ref_control)

eststo e_index: areg e_rf_index ref_info ref_cashonly, a(ref_stratum) cluster(village_id)
	qui sum e_rf_index if e(sample) == 1 & ref_control == 1
	estadd scalar control_mean = r(mean)
	
ritest rf_treatment _b[3.rf_treatment], r(2000) strata(county_name) fixlevels(2) cluster(village_id) seed(1): areg e_rf_index i.rf_treatment, a(ref_stratum)
	matrix p_info = r(p)
	
ritest rf_treatment _b[2.rf_treatment], r(2000) strata(county_name) fixlevels(3) cluster(village_id) seed(1): areg e_rf_index i.rf_treatment, a(ref_stratum)
	matrix p_cash = r(p)
	
matrix p_ri = (p_info , p_cash)
mat colnames p_ri = ref_info ref_cashonly
estadd matrix p_ri : e_index
matrix drop p_info p_cash p_ri

qui areg e_rf_index ref_info if ref_anycash == 1, a(ref_stratum) r
qui test ref_info = 0
estadd scalar p_label_grantonly = r(p) : e_index

	
forvalues i = 1/6{
eststo e_rf`i': areg e_rf_binary`i' ref_info ref_cashonly, a(ref_stratum) cluster(village_id)
	qui sum e_rf_binary`i' if e(sample) == 1 & ref_control == 1
	estadd scalar control_mean = r(mean)
	
ritest rf_treatment _b[3.rf_treatment], r(2000) strata(county_name) fixlevels(2) cluster(village_id) seed(1): areg e_rf_binary`i' i.rf_treatment, a(ref_stratum)
	matrix p_info = r(p)
	
ritest rf_treatment _b[2.rf_treatment], r(2000) strata(county_name) fixlevels(3) cluster(village_id) seed(1): areg e_rf_binary`i' i.rf_treatment, a(ref_stratum)
	matrix p_cash = r(p)
	
matrix p_ri = (p_info , p_cash)
mat colnames p_ri = ref_info ref_cashonly
estadd matrix p_ri : e_rf`i'	
matrix drop p_info p_cash p_ri

qui areg e_rf_binary`i' ref_info if ref_anycash == 1, a(ref_stratum) r
qui test ref_info = 0
estadd scalar p_label_grantonly = r(p) : e_rf`i'
}

** Impacts from second survey
foreach i in 1 2 3 4 5 6{

gen eb_rf_binary`i' = .

count if inlist(eb_rf`i',1,2)
local num_support = r(N)
count if inlist(eb_rf`i',4,5)
local num_oppose = r(N)

replace eb_rf_binary`i' = eb_rf`i' <= 3 if _merge_reffollowup == 3 & `num_support' < `num_oppose'
replace eb_rf_binary`i' = eb_rf`i' <= 2 if _merge_reffollowup == 3 & `num_support' >= `num_oppose'
}

swindex eb_rf_binary1 eb_rf_binary2 eb_rf_binary3 eb_rf_binary6, gen(eb_rf_index) replace normby(ref_cashonly)

eststo eb_index: areg eb_rf_index ref_info, a(ref_stratum) r
	qui sum eb_rf_index if e(sample) == 1 & ref_cashonly == 1
	estadd scalar cash_mean = r(mean)
	
forvalues i = 1/6{
eststo eb_rf`i': areg eb_rf_binary`i' ref_info, a(ref_stratum) r
	qui sum eb_rf_binary`i' if e(sample) == 1 & ref_cashonly == 1
	estadd scalar cash_mean = r(mean)
}

label var e_rf_index "\shortstack{Integration\\Policies\\Index}"
label var e_rf_binary1 "\shortstack{Supports\\Refugee\\Hosting}"
label var e_rf_binary6 "\shortstack{Supports\\More\\Refugees}"
label var e_rf_binary3 "\shortstack{Supports\\Right\\to Work}"
label var e_rf_binary2 "\shortstack{Supports\\Free\\Movement}"
label var e_rf_binary4 "\shortstack{Supports\\Providing\\Land}"
label var e_rf_binary5 "\shortstack{Supports\\Citizen\\-ship}"

esttab e_index e_rf1 e_rf6 e_rf3 e_rf2 e_rf4 e_rf5 using "$path/Output/relon.tex", label collabels(none) replace nolines nonumber keep(ref_info ref_cashonly) cells(b(star fmt(%9.2f)) se(par fmt(%9.2f)) p_ri(par([ ] ) fmt(%9.2f))) stats(N control_mean p_label_grantonly, fmt(%9.0fc %12.2f %9.2f) labels("Observations" "Control Mean" "Lab. Grant = Grant")) coeflabels(ref_info "Info. + Labeled Grant" ref_cashonly "Grant Only") order(ref_info ref_cashonly) substitute(\_ _ \$ $) star(* 0.1 ** 0.05 *** 0.01) ///
prehead("\begin{table}[h]	\centering	\footnotesize	\caption{Support for Refugee Integration Policies---Kenya} \label{tab:relon}	\begin{tabular}{l*{7}{>{\centering\arraybackslash}p{1.4cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-8} \textit{Immediate Impacts} &&&&&&& \\ ") ///
prefoot("&&&&&&& \\") ///
postfoot("")

esttab eb_index eb_rf1 eb_rf6 eb_rf3 eb_rf2 eb_rf4 eb_rf5 using "$path/Output/relon.tex", label collabels(none) append nolines nonumber nomtitle keep(ref_info) cells(b(star fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N cash_mean, fmt(%9.0fc %12.2f) labels("Observations" "Grant Only Mean")) coeflabels(ref_info "Info. + Labeled Grant") order(ref_info) substitute(\_ _ \$ $) star(* 0.1 ** 0.05 *** 0.01) ///
prehead("&&&&&&& \\") ///
posthead("\textit{Demand-Free Bound} &&&&&&& \\") ///
prefoot("&&&&&&& \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{8}{p{\linewidth}}{\footnotesize Each observation is a household in Kenya. \textit{Immediate Impacts} are measured the same day after grant and information distribution. \textit{Demand-Free Bound} computed using the method of \citet{de2018measuring} to identify the lower bound of demand-free treatment effects---labeled grant recipients receive a script attempting to induce negative demand effects, while grant only recipients receive a positive script. These results are measured using follow-up surveys conducted only in Labeled Grant and Grant Only about one month after the first survey (the omitted category is Grant Only). For \textit{Immediate Impacts} comparisons between Labeled Grant or Grant Only and Pure Control, standard errors are clustered at the village level and $ p $-values are computed through randomization inference, permuting treatment assignment 2,000 times using the Stata command \textit{ritest} \citep{ritest}. For comparisons between Labeled Grant and Grant Only, standard errors and $ p $-values are heteroskedasticity-robust. \textit{Lab. Grant = Grant} shows $ p $-values from a regression of the outcome on an \textit{Information +  Labeled Grant} indicator estimated on Information + Labeled Grant and Grant Only households only. Standard errors in parentheses; $ p $-values in brackets. \sym{*} \(p<0.1\), \sym{**} \(p<0.05\), \sym{***} \(p<0.01\).}  \end{tabular} \\ \end{table}%")

estimates drop e_index e_rf1 e_rf6 e_rf3 e_rf2 e_rf4 e_rf5 eb_index eb_rf1 eb_rf6 eb_rf3 eb_rf2 eb_rf4 eb_rf5

***** Table A7 -- Impacts from Kenya Extension, by Area Tercile of Control-Group Support

bys subcounty_name: egen subcounty_support = mean(e_rf_index) if ref_control == 1
gsort subcounty_name -ref_control
bys subcounty_name: replace subcounty_support = subcounty_support[1]

xtile area_support_tercile = subcounty_support, nq(3)

forvalues t=1/3{
eststo e_index_t`t': areg e_rf_index ref_info ref_cashonly if area_support_tercile == `t', a(ref_stratum) cluster(village_id)
	qui sum e_rf_index if e(sample) == 1 & ref_control == 1
	estadd scalar control_mean = r(mean)
	
ritest rf_treatment _b[3.rf_treatment], r(2000) fixlevels(2) cluster(village_id) seed(1): areg e_rf_index i.rf_treatment if area_support_tercile == `t', a(ref_stratum)
	matrix p_info = r(p)
	
ritest rf_treatment _b[2.rf_treatment], r(2000) fixlevels(3) cluster(village_id) seed(1): areg e_rf_index i.rf_treatment if area_support_tercile == `t', a(ref_stratum)
	matrix p_cash = r(p)
	
matrix p_ri = (p_info , p_cash)
mat colnames p_ri = ref_info ref_cashonly
estadd matrix p_ri : e_index_t`t'
matrix drop p_info p_cash p_ri

qui areg e_rf_index ref_info if ref_anycash == 1 & area_support_tercile == `t', a(ref_stratum) r
qui test ref_info = 0
estadd scalar p_label_grantonly = r(p) : e_index_t`t'

eststo eb_index_t`t': areg eb_rf_index ref_info if area_support_tercile == `t', a(ref_stratum) r
	qui sum eb_rf_index if e(sample) == 1 & ref_cashonly == 1
	estadd scalar cash_mean = r(mean)
}
	
esttab e_index_t1 e_index_t2 e_index_t3 using "$path/Output/Appendix_A/relon_subcounties.tex", nomtitle label collabels(none) replace nolines nonumber keep(ref_info ref_cashonly) cells(b(star fmt(%9.2f)) se(par fmt(%9.2f)) p_ri(par([ ] ) fmt(%9.2f))) stats(N control_mean p_label_grantonly, fmt(%9.0fc %12.2f %9.2f) labels("Observations" "Control Mean" "Labeled Grant = Grant")) coeflabels(ref_info "Info. + Labeled Grant" ref_cashonly "Grant Only") order(ref_info ref_cashonly) substitute(\_ _ \$ $) star(* 0.1 ** 0.05 *** 0.01) ///
prehead("\begin{table}[h]       \centering      \footnotesize   \caption{Support for Refugee Integration in Kenya, by Area Control-Group Support} \label{tab:relon_subcounties} \begin{tabular}{l*{3}{>{\centering\arraybackslash}p{2.2cm}}} \toprule \toprule & \multicolumn{3}{c}{Average Area Support:} \\ \cmidrule(l{4pt}r{4pt}){2-4} & Low & Medium & High \\") ///
posthead("\cmidrule{1-4} \textit{Immediate Impacts (Integration Support Index)} &&& \\ ") ///
prefoot("&&& \\") ///
postfoot("")

esttab eb_index_t1 eb_index_t2 eb_index_t3 using "$path/Output/Appendix_A/relon_subcounties.tex", label collabels(none) append nolines nonumber nomtitle keep(ref_info) cells(b(star fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N cash_mean, fmt(%9.0fc %12.2f) labels("Observations" "Grant Only Mean")) coeflabels(ref_info "Info. + Labeled Grant") order(ref_info) substitute(\_ _ \$ $) star(* 0.1 ** 0.05 *** 0.01) ///
prehead("&&& \\") ///
posthead("\textit{Demand-Free Bound (Integration Support Index)} &&& \\") ///
prefoot("&&& \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{4}{p{0.93\linewidth}}{\footnotesize See \autoref{tab:relon} for notes on estimation and sampling. Each observation is a household in Kenya. Sample is split into terciles based on the average value of the integration support summary index in the control group within the household's sub-county (there are 35 sub-counties). Two sub-counties without any Control observations are excluded from the sample. Standard errors in parentheses; $ p $-values in brackets. \sym{*} \(p<0.1\), \sym{**} \(p<0.05\), \sym{***} \(p<0.01\).}  \end{tabular} \\ \end{table}%")

estimates drop e_index_t1 e_index_t2 e_index_t3 eb_index_t1 eb_index_t2 eb_index_t3

***** Table A8 -- Impacts from Kenya Extension, by Area Tercile of Control-Group Consumption
gen resp_spend = e_c1*(30/7) + e_c2 + e_c4 + e_c5 + e_c6 + e_c7 + e_c8/12

	replace resp_spend = resp_spend / 132.50
winsor2 resp_spend, cuts(1 99) replace

bys subcounty_name: egen subcounty_spend = mean(resp_spend) if ref_control == 1
gsort subcounty_name -ref_control
bys subcounty_name: replace subcounty_spend = subcounty_spend[1]

xtile area_spend_tercile = subcounty_spend, nq(3)

forvalues t=1/3{
eststo e_index_t`t': areg e_rf_index ref_info ref_cashonly if area_spend_tercile == `t', a(ref_stratum) cluster(village_id)
	qui sum e_rf_index if e(sample) == 1 & ref_control == 1
	estadd scalar control_mean = r(mean)
	
ritest rf_treatment _b[3.rf_treatment], r(2000) fixlevels(2) cluster(village_id) seed(1): areg e_rf_index i.rf_treatment if area_spend_tercile == `t', a(ref_stratum)
	matrix p_info = r(p)
	
ritest rf_treatment _b[2.rf_treatment], r(2000) fixlevels(3) cluster(village_id) seed(1): areg e_rf_index i.rf_treatment if area_spend_tercile == `t', a(ref_stratum)
	matrix p_cash = r(p)
	
matrix p_ri = (p_info , p_cash)
mat colnames p_ri = ref_info ref_cashonly
estadd matrix p_ri : e_index_t`t'
matrix drop p_info p_cash p_ri

qui areg e_rf_index ref_info if ref_anycash == 1 & area_spend_tercile == `t', a(ref_stratum) r
qui test ref_info = 0
estadd scalar p_label_grantonly = r(p) : e_index_t`t'

eststo eb_index_t`t': areg eb_rf_index ref_info if area_spend_tercile == `t', a(ref_stratum) r
	qui sum eb_rf_index if e(sample) == 1 & ref_cashonly == 1
	estadd scalar cash_mean = r(mean)
}
	
esttab e_index_t1 e_index_t2 e_index_t3 using "$path/Output/Appendix_A/relon_subcounties_spend.tex", nomtitle label collabels(none) replace nolines nonumber keep(ref_info ref_cashonly) cells(b(star fmt(%9.2f)) se(par fmt(%9.2f)) p_ri(par([ ] ) fmt(%9.2f))) stats(N control_mean p_label_grantonly, fmt(%9.0fc %12.2f %9.2f) labels("Observations" "Control Mean" "Labeled Grant = Grant")) coeflabels(ref_info "Info. + Labeled Grant" ref_cashonly "Grant Only") order(ref_info ref_cashonly) substitute(\_ _ \$ $) star(* 0.1 ** 0.05 *** 0.01) ///
prehead("\begin{table}[h]       \centering      \footnotesize   \caption{Support for Refugee Integration in Kenya, by Area Control-Group Consumption} \label{tab:relon_subcounties_spend} \begin{tabular}{l*{3}{>{\centering\arraybackslash}p{2.2cm}}} \toprule \toprule & \multicolumn{3}{c}{Average Area Consumption:} \\ \cmidrule(l{4pt}r{4pt}){2-4} & Low & Medium & High \\") ///
posthead("\cmidrule{1-4} \textit{Immediate Impacts (Integration Support Index)} &&& \\ ") ///
prefoot("&&& \\") ///
postfoot("")

esttab eb_index_t1 eb_index_t2 eb_index_t3 using "$path/Output/Appendix_A/relon_subcounties_spend.tex", label collabels(none) append nolines nonumber nomtitle keep(ref_info) cells(b(star fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N cash_mean, fmt(%9.0fc %12.2f) labels("Observations" "Grant Only Mean")) coeflabels(ref_info "Info. + Labeled Grant") order(ref_info) substitute(\_ _ \$ $) star(* 0.1 ** 0.05 *** 0.01) ///
prehead("&&& \\") ///
posthead("\textit{Demand-Free Bound (Integration Support Index)} &&& \\") ///
prefoot("&&& \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{4}{p{0.93\linewidth}}{\footnotesize See \autoref{tab:relon} for notes on estimation and sampling. Each observation is a household in Kenya. Sample is split into terciles based on the average household consumption in the control group within the household's sub-county (there are 35 sub-counties). Two sub-counties without any Control observations are excluded from the sample. Standard errors in parentheses; $ p $-values in brackets. \sym{*} \(p<0.1\), \sym{**} \(p<0.05\), \sym{***} \(p<0.01\).}  \end{tabular} \\ \end{table}%")

estimates drop e_index_t1 e_index_t2 e_index_t3 eb_index_t1 eb_index_t2 eb_index_t3
drop resp_spend

** Table B5 - Randomization Balance, Kenya Extension

gen resp_id = .

forvalues i = 1/20 {
	replace resp_id = `i' if b_hr3 == "`i'" & e_rfresp1 == 1
	replace resp_id = `i' if b_hr4 == `i' & e_rfresp1 == 1
}
	replace resp_id = e_rfresp2 if e_rfresp1 == 0

gen resp_age = .
gen resp_female = .
gen resp_head = .
gen resp_edu = .
gen resp_married = .
gen resp_employed = .
gen resp_income = .
gen resp_hours_worked = .
gen resp_commute = .
gen resp_right_track = .
gen resp_happy = .

gen hhsize = 0

gen resp_spend = e_c1*(30/7) + e_c2 + e_c4 + e_c5 + e_c6 + e_c7 + e_c8/12
gen resp_savings = e_c9
gen resp_invest = (e_as1 + e_as2 + e_as3 + e_as4 + e_as5 + e_as6) / 12

forvalues i = 1/20 {
	
replace resp_age = b_hh1_`i' + 2 if resp_id == `i'
replace resp_female = b_hh2_`i' == 2 if b_hh2_`i' != . & resp_id == `i'
replace resp_head = b_hh3_`i' == 1 if b_hh3_`i' != . & resp_id == `i'
replace resp_edu = e_hh3b_`i' if resp_id == `i'
	replace resp_edu = e_hh3a_`i' if e_hh3a_`i' != . & resp_id == `i'
replace resp_married = e_m1_`i' == 2 if e_m1_`i' != . & resp_id == `i'
replace resp_employed = inlist(e_hh5_`i',1,2,3) if e_hh5_`i' != . & resp_id == `i'
replace resp_income = e_hh6_`i' if resp_id == `i'
replace resp_hours_worked = e_hh9_`i' if resp_id == `i'
	replace resp_hours_worked = 0 if e_hh9_`i' == . & resp_id == `i'
replace resp_commute = e_hh10_`i' if resp_id == `i'
	replace resp_commute = 0 if e_hh10_`i' == . & resp_id == `i'
replace resp_right_track = e_hh11_`i' <= 2 if e_hh11_`i' != . & resp_id == `i'	
replace resp_happy = e_hh12_`i' <= 2 if e_hh12_`i' != . & resp_id == `i'

replace hhsize = hhsize + 1 if e_hh1_`i' == 1
}

winsor2 resp_income resp_hours_worked resp_commute resp_spend resp_savings resp_invest, cuts(1 99) replace
	replace resp_income = resp_income / 132.50
	replace resp_spend = resp_spend / 132.50
	replace resp_savings = resp_savings / 132.50
	replace resp_invest = resp_invest / 132.50
	
label var resp_age "Age" 
label var resp_female "Female" 
label var resp_head "Head of Household" 
label var resp_edu "Education (Years)" 
label var resp_married "Married" 
label var resp_employed "Employed" 
label var resp_income "Income"
label var resp_hours_worked "Hours Worked, Past Week"
label var resp_commute "Commute Time, Minutes"
label var resp_right_track "Life on Right Track"
label var resp_happy "Mostly Happy, Past Month"
label var hhsize "Household Size" 
label var resp_spend "Household Expenditure" 
label var resp_savings "Household Savings" 
label var resp_invest "Household Durable Investment"

local balance_vars resp_age resp_female resp_head resp_edu resp_married resp_employed resp_income resp_hours_worked resp_commute resp_right_track resp_happy hhsize resp_spend resp_savings resp_invest

eststo tb6_1: estpost summarize `balance_vars' if rf_treatment == 1
eststo tb6_2: estpost summarize `balance_vars' if rf_treatment == 2
eststo tb6_3: estpost summarize `balance_vars' if rf_treatment == 3
eststo p_joint: reg rf_treatment `balance_vars' // to get column titles right

local iter = 0
foreach var in `balance_vars' {
	local++ iter
	qui areg `var' i.rf_treatment, cluster(village_id) a(ref_stratum)
	matrix N_`iter' = e(N)
	test 2.rf_treatment = 3.rf_treatment = 0
	matrix p_`iter' = r(p)
}

matrix p_joint = (p_1 , p_2, p_3, p_4, p_5, p_6, p_7, p_8, p_9, p_10, p_11, p_12, p_13, p_14, p_15)
mat colnames p_joint = `balance_vars'
estadd matrix p_joint : p_joint
matrix N_joint = (N_1 , N_2, N_3, N_4, N_5, N_6, N_7, N_8, N_9, N_10, N_11, N_12, N_13, N_14, N_15)
mat colnames N_joint = `balance_vars'
estadd matrix N_joint : p_joint
matrix drop p_joint p_1 p_2 p_3 p_4 p_5 p_6 p_7 p_8 p_9 p_10 p_11 p_12 p_13 p_14 p_15 N_1 N_2 N_3 N_4 N_5 N_6 N_7 N_8 N_9 N_10 N_11 N_12 N_13 N_14 N_15

eststo tb6_1_f: estpost summarize `balance_vars' if rf_treatment == 1
eststo tb6_2_f: estpost summarize `balance_vars' if rf_treatment == 2
eststo tb6_3_f: estpost summarize `balance_vars' if rf_treatment == 3
eststo p_joint_f: reg rf_treatment `balance_vars' // to get column titles right

local iter = 0
foreach var in `balance_vars' {
	local++ iter
	qui areg `var' i.rf_treatment if _merge_reffollowup == 3, r a(ref_stratum)
	matrix N_`iter' = e(N)
	test 2.rf_treatment = 3.rf_treatment = 0
	matrix p_`iter' = r(p)
}

matrix p_joint = (p_1 , p_2, p_3, p_4, p_5, p_6, p_7, p_8, p_9, p_10, p_11, p_12, p_13, p_14, p_15)
mat colnames p_joint = `balance_vars'
estadd matrix p_joint : p_joint_f
matrix N_joint = (N_1 , N_2, N_3, N_4, N_5, N_6, N_7, N_8, N_9, N_10, N_11, N_12, N_13, N_14, N_15)
mat colnames N_joint = `balance_vars'
estadd matrix N_joint : p_joint_f
matrix drop p_joint p_1 p_2 p_3 p_4 p_5 p_6 p_7 p_8 p_9 p_10 p_11 p_12 p_13 p_14 p_15 N_1 N_2 N_3 N_4 N_5 N_6 N_7 N_8 N_9 N_10 N_11 N_12 N_13 N_14 N_15

esttab tb6_1 tb6_2 tb6_3 p_joint using "$path/Output/Appendix_B/relon_balance.tex", keep(`balance_vars') order(`balance_vars') label nostar nomtitles substitute(\_ _) replace collabels(none) nolines nonumber noobs cells("mean(pattern(1 1 1 0) fmt(a2)) p_joint(pattern(0 0 0 1) fmt(2)) N_joint(pattern(0 0 0 1) fmt(%9.0fc))") ///
prehead("\begin{table}[H]	\centering	\footnotesize	\caption{Randomization Balance, Kenya Extension} \label{tab:relon_balance} \begin{tabular}{l*{5}{>{\centering\arraybackslash}p{1.5cm}}} \toprule \toprule  & \shortstack{Mean:\\Pure\\Control} & \shortstack{Mean:\\Grant\\Only} & \shortstack{Mean:\\Labeled\\Grant} & \shortstack{Joint\\ $ p $-Value\\~} & \shortstack{N\\~\\~} \\") ///
posthead("\cmidrule{2-6}    \textit{Full Sample} &       &       &       &  \\") ///
prefoot("&&&&& \\") postfoot("")

esttab tb6_1_f tb6_2_f tb6_3_f p_joint_f using "$path/Output/Appendix_B/relon_balance.tex", keep(`balance_vars') order(`balance_vars') label nostar nomtitles substitute(\_ _) append collabels(none) nolines nonumber noobs cells("mean(pattern(1 1 1 0) fmt(a2)) p_joint(pattern(0 0 0 1) fmt(2)) N_joint(pattern(0 0 0 1) fmt(%9.0fc))" ) ///
prehead("") ///
posthead("  \textit{Follow-Up Sample} &       &       &       &  \\") ///
prefoot("") ///
postfoot("\bottomrule \bottomrule \multicolumn{6}{p{0.87\linewidth}}{\footnotesize \textit{Follow-Up Sample} includes households surveyed at the one-month follow-up in Kenya; the follow-up was not conducted with the pure control group. First three columns show means within treatment groups. Fourth column shows $ p $-values from joint F-tests that means are equal in all treatment groups, recovered from a regression of each variable on treatment and randomization-stratum dummies with standard errors that are clustered at the village level in the full sample and heteroskedasticity-robust in the follow-up sample. Monetary units are USD/month.}  \end{tabular} \\ \end{table}%")

estimates drop tb6_1 tb6_2 tb6_3 p_joint tb6_1_f tb6_2_f tb6_3_f p_joint_f

** Table C3 - Expectations of Future Assistance, Kenya Extension

forvalues i=1/2{
eststo tc3_`i': areg eb_ge`i' ref_info, a(ref_stratum) r
	qui sum eb_ge`i' if e(sample) == 1 & ref_cashonly == 1
	estadd scalar cash_mean = r(mean)
}

label var eb_ge1 "\shortstack{Expects Aid From\\Within Village}"
label var eb_ge2 "\shortstack{Expects Aid From\\Outside Village}"

esttab tc3_1 tc3_2 using "$path/Output/Appendix_C/aid_expectations.tex", label collabels(none) replace nolines nonumber keep(ref_info) cells(b(star fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N cash_mean, fmt(%9.0fc %12.2f) labels("Observations" "Control Mean")) coeflabels(ref_info "Info. + Labeled Grant") order(ref_info) substitute(\_ _ \$ $) star(* 0.1 ** 0.05 *** 0.01) ///
prehead("\begin{table}[h]	\centering	\footnotesize	\caption{Expectations of Future Aid---Kenya} \label{tab:aid_expectations}	\begin{tabular}{l*{2}{>{\centering\arraybackslash}p{4cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-3}") ///
prefoot("&& \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{3}{p{0.68\linewidth}}{\footnotesize Each observation is a household. These results are measured using follow-up surveys conducted only in Labeled Grant and Grant Only about one month after the first survey in Kenya. Outcomes are measured using survey questions asking whether the respondent expects to receive cash gifts from anyone inside (outside) their village in the next 3 months. Heteroskedasticity-robust standard errors in parentheses; $ p $-values in brackets. \sym{*} \(p<0.1\), \sym{**} \(p<0.05\), \sym{***} \(p<0.01\).}  \end{tabular} \\ \end{table}%")

estimates drop tc3_1 tc3_2
